Problem:
Score : 700 points

Problem Statement

Takahashi is locked within a building.

This building consists of H×W rooms, arranged in H rows and W columns.
We will denote the room at the i-th row and j-th column as (i,j). The state of this room is represented by a character A_{i,j}. If A_{i,j}= #, the room is locked and cannot be entered; if A_{i,j}= ., the room is not locked and can be freely entered.
Takahashi is currently at the room where A_{i,j}= S, which can also be freely entered.

Each room in the 1-st row, 1-st column, H-th row or W-th column, has an exit.
Each of the other rooms (i,j) is connected to four rooms: (i-1,j), (i+1,j), (i,j-1) and (i,j+1).

Takahashi will use his magic to get out of the building. In one cast, he can do the following:

Move to an adjacent room at most K times, possibly zero. Here, locked rooms cannot be entered.

Then, select and unlock at most K locked rooms, possibly zero. Those rooms will remain unlocked from then on.

His objective is to reach a room with an exit. Find the minimum necessary number of casts to do so.

It is guaranteed that Takahashi is initially at a room without an exit.

Constraints

3 ≤ H ≤ 800

3 ≤ W ≤ 800

1 ≤ K ≤ H×W

Each A_{i,j} is # , . or S.

There uniquely exists (i,j) such that A_{i,j}= S, and it satisfies 2 ≤ i ≤ H-1 and 2 ≤ j ≤ W-1.

Input

Input is given from Standard Input in the following format:

H W K
A_{1,1}A_{1,2}...A_{1,W}
:
A_{H,1}A_{H,2}...A_{H,W}

Output

Print the minimum necessary number of casts.

Sample Input 1

3 3 3
#.#
#S.
###

Sample Output 1

1

Takahashi can move to room (1,2) in one cast.

Sample Input 2

3 3 3
###
#S#
###

Sample Output 2

2

Takahashi cannot move in the first cast, but can unlock room (1,2).
Then, he can move to room (1,2) in the next cast, achieving the objective in two casts.

Sample Input 3

7 7 2
#######
#######
##...##
###S###
##.#.##
###.###
#######

Sample Output 3

2

Solution in Pascal (FPC 2.6.2) (judge verdict: Accepted)

var
  n,m,k,i,j,x,y,head,tail,tot,ans:longint;
  map:array[0..801,0..801]of char;
  cost,dx,dy:array[0..650000]of longint;
  p:array[0..801,0..801]of boolean;
  s:ansistring;

function min(x,y:longint):longint;
begin
    if x<y then exit(x) else exit(y);
end;

begin
    readln(n,m,k);
    for i:=1 to n do begin
        readln(s);
        for j:=1 to m do begin
            map[i,j]:=s[j];
            if map[i,j]='S' then begin
                x:=i;
                y:=j;
                if (x=1)or(x=n)or(y=1)or(y=n) then begin
                    writeln(0);
                    halt;
                end;
            end;
            if map[i,j]='.' then p[i,j]:=true;
        end;
    end;
    head:=1; tail:=1;
    dx[1]:=x; dy[1]:=y;
    while head<=tail do begin
        x:=dx[head]; y:=dy[head];
        if cost[head]=k+1 then break;
        if p[x-1,y] and (x>1) then begin
            inc(tail);
            dx[tail]:=x-1; dy[tail]:=y;
            cost[tail]:=cost[head]+1;
            p[x-1,y]:=false;
        end;
        if p[x+1,y] and (x<n) then begin
            inc(tail);
            dx[tail]:=x+1; dy[tail]:=y;
            cost[tail]:=cost[head]+1;
            p[x+1,y]:=false;
        end;
        if p[x,y-1] and (y>1) then begin
            inc(tail);
            dx[tail]:=x; dy[tail]:=y-1;
            cost[tail]:=cost[head]+1;
            p[x,y-1]:=false;
        end;
        if p[x,y+1] and (y<m) then begin
            inc(tail);
            dx[tail]:=x; dy[tail]:=y+1;
            cost[tail]:=cost[head]+1;
            p[x,y+1]:=false;
        end;
        inc(head);
    end;
    ans:=maxlongint;
    for i:=1 to head-1 do begin
        x:=dx[i]; y:=dy[i];
        tot:=min(min(x-1,y-1),min(n-x,m-y));
        if tot mod k=0 then ans:=min(ans,tot div k)
        else ans:=min(ans,(tot div k)+1);
    end;
    writeln(ans+1);
end.